Всё о файлах (Статья)

Всё о файлах (Статья)
Решил здесь собрать воедино основные приемы работы с файлами.
== Текстовые файлы ==
Текстовый файл отличается тем что он разбит на разные по длине строки, отделенные символами #13#10. Есть 2 основных метода работы с текстовыми файлами - старый паскалевский способ и через файловые потоки. У обоих есть преимущества и недостатки. Через потоки способ проще поэтому начнем с него.
Итак у всех потомков класса TStrings (TStringList, memo.Lines и т.п. ) есть методы записи и чтения в файл - SaveToFile, LoadFromFile. Преимущество - простота использования и довольно высокая скорость, недостаток - читать и писать файл можно только целиком.
Примеры.
1) Загрузка текста из файла в Memo:
Memo1.Lines.LoadFromFile('c:\MyFile.txt');
2) Сохранение в файл:
Memo1.Lines.SaveToFile('c:\MyFile.txt');
3) А вот так можно прочитать весь файл в строку:
function ReadFromFile(FileName: string): string;
begin
 with TStringList.Create do
 try
 LoadFromFile(FileName);
 Result := Text;
 finally
 Free;
 end;
end;
== Паскалевский метод доступа ==
Для более тонких операций над текстовыми файлами придется освоить очень древний паскалевский способ.
Итак, для доступа к текстовым файлам используется переменная типа TextFile. До сих пор не совсем понимаю что это такое физически - что-то типа "внутреннего" паскалевского Handle на файл.
Итак чтобы ассоциировать файл на диске с переменной надо проделать следующие опрерации:
1) Определяем файловую переменную:
var f: TextFile;
2) Ассоциируем ее:
AssignFile(F, 'c:\MyFile.txt');
3) Теперь надо этот файл открыть, есть 3 варианта:
* Rewrite(f) - файла нет или он должен быть перезаписан, открытие для записи
* Reset(f) - файл есть и его надо открыть для чтения (с первой строки)
* Append(f) - файл есть и его надо открыть для дописования строк в конец
Как видите не хватает очень полезных функций таких как открытия файла для чтения с произвольной строки и для записи в файл произвольной строки. Но надо учесть, что так как длины строк разные, не существует никакого способа узнать физическое место начала например 1000 строки, не прочитав всю тысячу строк. Для записи ситуация еще сложнее - вставить строку означает перезаписать всю информацию после этой строки заново. Таким образом варианты только следующие:
* Перезаписать весть файл
* Читать с первой строки
* Дописать что-то в конец
* Читать и писать файл целиком (см. выше работу через TStrings)
В конце работы открытый файл нужно закрыть:
CloseFile(f);
Теперь пусть у нас есть строковая переменная s для чтения строки из файла
Чтение предварительно открытого файла:
ReadLn(f, s) - будет прочитанна текущая строка и позиция чтения переведена на следующую позицию.
А как прочитать весь файл?
while not Eof(f) do
begin
 ReadLn(f, s);
 { здесь делаем что-то с прочитанной строкой }
end;
Хорошо, а если файл несколько метров есть ли способ поставить какой-нибудь ProgressBar или Gauge чтобы показывал сколько считано? Есть, но не совсем прямой - не забыли, сколько строк в файле заранее мы не знаем, узнать можно только прочитав его весь, но показометер мы все-таки сделаем:
var
 Canceled: Boolean;
function GetFileSize(FileName: string): Integer;
var
 f: File of Byte;
begin
 try
 AssignFile(f, FileName);
 Reset(f);
 Result := FileSize(F);
 CloseFile(f);
 except
 Result := -1;
 end;
end;
procedure ReadMyFile();
var
 i, j: Integer;
begin
 ProgressBar1.Max := GetFileSize('c:\MyFile.txt');
 ProgressBar1.Position := 0;
 AssignFile(f, 'c:\MyFile.txt');
 Canceled := False;
 Reset(f);
 i := 0;
 j := 0;
 while not Eof(f) do
 begin
 Inc(j);
 ReadLn(f, s);
 i := i + Length(s) + 2;
 if (j mod 1000) = 0 then
 begin
 ProgressBar1.Position := i;
 Application.ProcessMessages;
 if Canceled then break;
 end;
 { здесь мы что-то делаем с прочитанной строкой }
 end;
 CloseFile(f);
end;
Теперь комментарии к коду.
1) Функцию GetFileSize я рассмотрю после, она немного по другому подходит к чтению файла (кстати я знаю еще по крайней мере 3 способа ее реализации, поэтому не нужно указывать что это можно сделать легче, быстрее или просто по другому - просто давайте разберем это позже)
2) Переменная i - все время указывает на количество байт которое мы считали - мы определяем длину каждой строки и прибавляем 2 (символы конца строки). Зная длину файла в байтах и сколько байт прочитано можно оценить и прогресс, но
3) Если ставить изменение прогресса после каждой строки, то это очень сильно тормознет процесс. Поэтому вводим переменную j и обновляем прогресс например 1 раз на 1000 прочитанных строк
4) Переменная Canceled - глобальная переменная. Поставьте на форму кнопку, в обработчике нажатия поставьте Canceled := True; и нажатие кнопки прервет чтение файла.
Теперь как писать в текстовый файл:
* Writeln(f, s) - Запись целой строки
* Write(f, s) - Запись кусочка строки(те следующая операция записи будет произведена в ту же строку)
Если переменная s содержит больше 255 символов (т.е. является длинной строкой), то таким способом ни фига не запишится, в файл вместо строки попадут 4 байта указателя на нее. Надо делать так:
WriteLn(f, PChar(s)^);
== Типизированные файлы ==
Теперь разберем типизированные файлы. Типизированный файл - это файл в котором записанны идентичные структуры. Например любой файл можно считать файлом байтов - т.е. можно его читать байт за байтом, можно перейти сразу к любому байту по его номеру, можно сразу узнать сколько байт в файле, можно заменить любой байт на другой не перезаписывая файл. Теперь все это в примерах:
// Объявляем файл байтов:
var
 f: file of Byte;
 b: Byte;
 
Ассоциируем файловую переменную с физическим файлом: AssignFile(f, 'c:\myfile.bin');
Теперь мы можем либо перезаписать/создать файл: Rewrite(f);
 
Либо открыть существующий для чтения и записи: Reset(f);
Обратите внимание, что функция Reset хотя и имеет такой же формат как и для текстовых файлов, но открытый ей файл можно и читать и писать, в отличие от текстовых.
Теперь функции работы с файлом:
* Read(f, b) - прочитать 1 байт
* Write(f, b) - записать 1 байт
* Seek(f, 100) - поставить текущее положение считывания/записи на сотый байт
* Size(f) - прочитать количество байт в файле.
* Eof(f) - узнать не являетсмя ли байт последним
Все эти функции не работают с файлами большими 2 Gb.
После работы файл надо закрыть: CloseFile(f);
Приведенные выше механизмы будут работать с любым файлом, так как любой файл можно считать файлом байтов. Теперь где это можно использовать? В принципе везде, но в подавляющем большинстве случаев это будет очень неудобно, ведь скорость считывания при чтении по байтам будет на порядки более низкой чем другими способами. Однако в некоторых случаях этот способ может быть очень полезен. Например в программе вам надо заменить 100й байт файла на другой, или прочитать 100й байт файла, например во всяких читерских программах, при взломе и т.п. Здесь такой доступ будет весьма удобен. Гораздо более интересным представляется дальнейшее развитие технологии типизированных файлов (их еще лет 15 назад называли "Файлы прямого доступа"). Представим себе, что файл состоит не из байт а из более сложных структур. Например мы имеем некоторую информацию в виде:
type MyRec = record
 Name: string[100];
 Age: Byte;
 Membership: Boolean;
 Accounts: array[1..10] of Integer;
end;
Обратите внимание, что все элементы записи точно типизированы, нет ни длинных строк, ни открытых массивов, ни объектов, другими словами, заранее точно известно сколько именно байт будет занимать переменная этого типа. Объявим переменную этого типа:
var MyVar: MyRec;
и файл этого типа:
var f: file of MyRec;
Теперь мы можем читать и писать сразу целую структуру, абсолютно так же как и если бы это был один байт:
AssignFile(f, 'c:\MyFile.rec');
Rewrite(f);
MyVar.Name := 'Vitaly';
MyVar.Age := 33;
MyVar.Membership := True;
MyVar.Accounts[1] := 12345;
MyVar.Accounts[2] := 34985;
Write(f, MyVar);
Closefile(f);
Все остальные функции приведенные в предыдущей статье будут работать так же, только одно отличие - Seek и Size оперируют не с количеством байт, а с количеством записей.
== Нетипизированные файлы ==
Идем дальше. Есть такое понятие как нетипизированный файл. Это такой файл который содержит разнородные элементы. Например файл EXE - вначале он имеет заголовок, затем двоичный код, в конце какие-то ресурсы. Все части файла имеют разную длину и разную структуру. Тут уже обратится к произвольному элементу сложно, обычно надо вначале узнать где этот элемент находится, подчас это записано в предыдущем куске информации. Работа с такими файлами достаточно сложна и требует вручную разработки алгоритмов его чтения, но в связи гибкостью структуры и компактностью такие файлы составляют большинство. Для работы с нетипизированными файлами используют процедуры BlockRead и BlockWrite, которые позволяют читать/писать произвольное количество байт. Привожу пример пользования этими функциями из справки по Дельфи:
var
 FromF, ToF: file;
 NumRead, NumWritten: Integer;
 Buf: array[1..2048] of Char;
begin
 if OpenDialog1.Execute then { Display Open dialog box }
 begin
 AssignFile(FromF, OpenDialog1.FileName);
 Reset(FromF, 1); { Record size = 1 }
 if SaveDialog1.Execute then { Display Save dialog box}
 begin
 AssignFile(ToF, SaveDialog1.FileName); { Open output file }
 Rewrite(ToF, 1); { Record size = 1 }
 Canvas.TextOut(10, 10, 'Copying ' + IntToStr(FileSize(FromF))+ ' bytes...');
 repeat
 BlockRead(FromF, Buf, SizeOf(Buf), NumRead);
 BlockWrite(ToF, Buf, NumRead, NumWritten);
 until (NumRead = 0) or (NumWritten <> NumRead);
 CloseFile(FromF);
 CloseFile(ToF);
 end;
 end;
end;
Этот код копирует из одного файла в другой. Замечания по поводу этого метода работы с файлами - плюсы - очень высокая скорость, особенно если размер буффера увеличить до 64kb-512kb, что позволит считывать файл достаточно большими кусками, чтобы обеспечить отсутствие простоев винчестера, к тому же обеспечивается очень высокая гибкость в работе. Минусы - сложность разработки, необходимость вручную писать все детали механизма чтения/записи и интерпретации данных.
Пожалуй на этом можно было бы и завершить описание работы с файлами средствами Паскаля и файловых переменных, но заглянув в Help Дельфей я обнаружил еще несколько функций достойных упоминания.
* Erase(f) - удаляет файл
* FilePos(f) - возвращает текущую позицию чтения/записи в файл
* Flush(f) - сбрасывает кэшированные файловые операции на диск
* Rename(f, 'MyNewFileName.txt') - переименование файлов
* Truncate(f) - файл обрезается до текущей позиции чтения/записи
== Файловые потоки ==
Теперь разберем возможности работы потомка TStream - TFileStream - файловый поток. Этот класс был специально введен для работы с файлами. Для работы с файловым потоком Вам надо записать в Uses модули classes, Sysutils (classes - включает в себя собственно определение класса, Sysutils - некоторые константы необходимые для работы).
Вот пример записи/перезаписи файла:
procedure WriteFileUsingStream(s, FileName: string);
begin
 with TFileStream.Create(FileName, fmCreate or fmOpenWrite) do
 try
 Write(PChar(s)^, Length(s));
 finally
 Free;
 end;
end;
Теперь небольшой разбор:
TFileStream.Create - конструктор класса, его вызов требует указания имени файла и опций его открытия, следующие опции определены:
 fmCreate = $FFFF;
 fmOpenRead = $0000;
 fmOpenWrite = $0001;
 fmOpenReadWrite = $0002;
 fmShareCompat = $0000;
 fmShareExclusive = $0010;
 fmShareDenyWrite = $0020;
 fmShareDenyRead = $0030;
 fmShareDenyNone = $0040;
Теперь метод Write - этим методом в файл пишется любая информация из буфера любого типа, Вам надо указать только буфер и количество записываемых байтов. В данном случае используется переменная типа String в качестве буффера, но так как для длинных строк она представляет собой лишь указатель, то конструкция "PChar(s)^" заставляет обращаться именно к ее содержимому.
А вот этот код демонстрирует чтение файла с использованием файлового потока:
var 
 p: PChar;
begin
 GetMem(p, 255);
 with TFileStream.create('c:\myText.txt', fmOpenReadWrite) do
 try
 Seek(10, soFromBeginning);
 Read(p^, 254);
 finally
 Free;
 end;
 ShowMessage(p);
 FreeMem(p);
end;
И пояснения к коду:
1) Никаких проверок длину файла и его наличие здесь не делается - это демонстрационный код, а не готовая процедура чтения.
2) Файл мы считываем в буфер типа PChar (с тем же успехом можно использовать массив или любой другой контейнер). Для тех кто не помнит - процедуры GetMem(p, 255) и FreeMem(p) - распределение памяти для строки и освобождение памяти.
3) Метод потока Seek позволяет установить текущую позицию считывания/записи файла. Первый параметр - номер байта, второй - это от чего считать этот байт (у нас считать от начала файла), возможны варианты:
* soFromBeginning - от начала файла
* soFromCurrent - от текущей позиции считывания
* soFromEnd - от конца файла (в этом случае номер байта должен быть отрицательным или равным нулю)
4) Собственно считывание из потока осуществляется методом read, в котором указывается в качестве параметров буфер в который мы читаем и желаемое количество байт для чтения. Метод read является функцией, которая возвращает количество байт реально прочитанных из потока.
Заканчивая о файловых потоках хочу упомянуть о методе CopyFrom который позволяет перекачивать информацию из одного потока в другой и о свойствах:
* Size - размер файла
* Position - текущая позиция чтения/записи потока
Работа с файловыми потоками весьма быстра, этот класс, являсь классом VCL, в то же время базируется на низкоуровневых функциях Windows, что обеспечивает очень высокую скорость работы и стабильность операций. К тому же многие компоненты и классы VCL поддерживаю прямое чтение и запись с файловыми потоками, что занчительно упрощает работу - например TStringList, TBlobField, TMemoField и другие.
Файловые потоки могут быть рекомендованы к использованию в большинстве случаев для чтения и записи файлов (за исключением специфических ситуаций, требующих каких-то других подходов), другими словами если вам надо просто записать или считать файл, используйте файловые потоки.
== Работа через Handle ==
Еще один способ работы с файлами - это открытие Handle на файл и работу через него. Тут есть 2 варианта - можно использовать функции Delphi или использовать WinAPI напрямую.
При использовании функций Delphi можно применять следующие функции:
* function FileOpen(FileName, fmOpenWrite or fmShareDenyNone): Integer - функция открывает файл и возвращает целое число - Handle на файл. Параметры функции - имя файла и тип доступа (все типы доступа я перечислил ранее). Если файл успешно открыт то Handle должен быть положительным числом, отрицательное число - это код ошибки.
Во всех остальных функциях используется именно значение Handle, возвращаемое этой функцией.
* procedure FileClose(Handle: Integer) - закрывает файл
function FileRead(Handle: Integer; var Buffer; Count: Integer): Integer;
function FileWrite(Handle: Integer; const Buffer; Count: Integer): Integer;
Эти функции для чтения/записи файла, где Buffer любая переменная достаточного размера для чтения/записи куска информации (обычно типа PChar или массив), Count-количество байт, которое Вы желаете записать/прочитать. Функции возвращают количество байт которые реально были прочитаны или записаны.
Этот тип доступа к файлам применяется весьма редко. Дело в том что он практически дублирует соответствующие функции WinAPI и к тому же обычно работает несколько медленнее, чем например потоки. И все же использование функций FileOpen и FileClose не лишено привлекательности. Наряду с тем что эти функции намного легче в использовании соответствующих функций WinAPI (можете сравнить - FileOpen имеет 2 параметра, cooтветствующая функция WinAPI - CreateFile имеет 7 параметров, большая часть из которых реально требуется лишь в ограниченном числе случаев) этот путь доступа открывает возможность прямого использования всех функций WinAPI про работе с файлами, которые требуют Handle на открытый файл.
== Файловые операции ==
Delphi предоставляет довольно широкие возможности по файловым операциям без использования механизмов открытия/закрытия файлов.
Вот список наиболее употребимых функций, большинство из которых в качестве параметров нуждаются только в имени файла:
* procedure ChDir(NewCurrentPath: string); - изменяет текущий каталог (в среде Windows сие конечно не так актуально как в ДОС, но все же), прочитать же текущий каталог можно функцией GetCurrentDir, а текущий каталог для определенного драйва - GetDir.
* function CreateDir(const Dir: string): Boolean; - создает каталог. При этом предыдущий уровень должен присутствовать. Если вы хотите сразу создать всю вложенность каталогов используйте функцию function ForceDirectories(Dir: string): Boolean; Обе функции возвращают True если каталог создан
* function DiskFree(Drive: Byte): Int64; - дает свободное место на диске. Параметер - номер диска 0 = текущий, 1 = A, 2 = B, и так далее
* function DiskSize(Drive: Byte): Int64; - размер винта. Обратите внимание на то что для результата этой и предыдущей функций абсолютно необходимо использовать переменную типа Int64, иначе максимум того что вы сможете прочитать правильно будет ограничен 2Gb
* function FileExists(const FileName: string): Boolean; - применяется для проверки наличия файла
* function FileGetAttr(const FileName: string): Integer;
* function FileSetAttr(const FileName: string; Attr: Integer): Integer; - функции для работы с атрибутами файлов. Вот список возможных атрибутов:
 faReadOnly $00000001 Read-only files
 faHidden $00000002 Hidden files
 faSysFile $00000004 System files
 faVolumeID $00000008 Volume ID files
 faDirectory $00000010 Directory files
 faArchive $00000020 Archive files
 faAnyFile $0000003F Any file
(Естественно не все атрибуты применимы во всех случаях)
* function RemoveDir(const Dir: string): Boolean; - удаляет папку(пустую)
* function DeleteFile(const FileName: string): Boolean; - удаляет файл
* procedure RenameFile(const OldName, NewName: string) - переименовывает файл
== Информация о файле ==
Привожу пример функции которая собирает довольно большое количество информации о выбранном файле:
type TFileInfo = record
 Exists: Boolean; // True если файл найден
 Name: string; // имя файла с расширением
 ShortName: string; // DOS 8.3 имя файла
 NameNoExt: string; // имя файла без расширения
 Extension: string; // расширение файла
 AssociatedFile: string; // программа с которой ассоциирован файл
 Path: string; // путь к файлу
 ShortPath: string; // DOS 8.3 путь файла
 Drive: string; // дисковод на котором находится файл
 CreateDate: TDateTime; // время когда файл создан
 Size: Int64; // размер файла (работает для файлов и больше 2Gb)
 Attributes: record // наличие/отсутствие системных атрибутов
 ReadOnly: Boolean;
 Hidden: Boolean;
 System: Boolean;
 Archive: Boolean;
 end;
 ModifyDate: TDateTime; // время последнего изменения файла
 LastAccessDate: TDateTime; // дата последнего открытия файла
end;
function ReadFileInfo(FileName: string): TFileInfo;
var
 ts: TSearchRec;
 function FileTime2DateTime(FT: _FileTime): TDateTime;
 var
 FileTime: _SystemTime;
 begin
 FileTimeToLocalFileTime(FT, FT);
 FileTimeToSystemTime(FT, FileTime);
 Result := EncodeDate(FileTime.wYear, FileTime.wMonth, FileTime.wDay)
 + EncodeTime(FileTime.wHour, FileTime.wMinute, FileTime.wSecond, FileTime.wMilliseconds);
 end;
 function AssociatedFile(FileExt: string): string;
 var
 key: string;
 begin
 with TRegistry.Create do
 try
 RootKey := HKEY_CLASSES_ROOT;
 OpenKey(FileExt, False);
 Key := ReadString();
 CloseKey;
 OpenKey(key + '\Shell\open\command', False);
 Result := ReadString();
 Closekey;
 finally
 Free;
 end;
 end;
begin
 Result.Name := ExtractFileName(FileName);
 Result.Extension := ExtractFileExt(FileName);
 Result.NameNoExt := Copy(Result.Name, 1, Length(Result.Name) - Length(Result.Extension));
 Result.Path := ExtractFilePath(FileName);
 Result.Drive := ExtractFileDrive(FileName);
 Result.ShortPath := ExtractShortPathName(ExtractFilePath(FileName));
 if LowerCase(Result.Extension) <> '.exe' then
 Result.AssociatedFile := AssociatedFile(Result.Extension);
 if FindFirst(FileName, faAnyFile, ts) = 0 then
 begin
 Result.Exists := True;
 Result.CreateDate := FileDateToDateTime(ts.Time);
 Result.Size := ts.FindData.nFileSizeHigh * 4294967296 + ts.FindData.nFileSizeLow;
 Result.Attributes.ReadOnly := ((faReadOnly and ts.Attr) > 0);
 Result.Attributes.Hidden := ((faHidden and ts.Attr) > 0);
 Result.Attributes.System := ((faSysFile and ts.Attr) > 0);
 Result.Attributes.Archive := ((faArchive and ts.Attr) > 0);
 Result.ModifyDate := FileTime2DateTime(ts.FindData.ftLastWriteTime);
 Result.LastAccessDate := FileTime2DateTime(ts.FindData.ftLastAccessTime);
 Result.ShortName := ts.FindData.cAlternateFileName;
 Findclose(ts);
 end
 else
 Result.Exists := False;
end;
Скорее всего эта функция как есть вряд ли понадобится, так как наверняка бОльшее количество определяемых параметров избыточно, тем ни менее может кому пригодится как пример выяснения информации о файле.
== Поиск файлов ==
Теперь поговорим о поиске файлов. Для этой цели могут использоваться процедуры FindFirst, FindNext, FindClose, при участии переменной типа TSearchRec которая хранит информацию о текущем статусе поиска и характеристики последнего найденного файла.
Пример иллюстрирующий поиск всех файлов и каталогов в определенном каталоге:
var
 SearchRec:TSearchRec;
begin
 if FindFirst('c:\Windows\*.*', faAnyFile, SearchRec) = 0 then
 repeat
 { Вот здесь мы можем делать с найденным файлом что угодно
 SearchRec.name - имя файла
 ExpandFileName(SearchRec.name) - имя файла с полным путем}
 until FindNext(SearchRec) <> 0;
 FindClose(SearchRec);
end;
Примечания по приведенному коду:
1) Первыми в список могут попадать файлы с именами "." и ".." - это ДОСовские имена для переходов на "родительский уровень", иногда нужна обработка для их игнорирования.
2) FindFirst в качестве первого параметра принимает шаблон для поиска, так как он был принят для ДОС. Если шаблон не включает путь то файлы будут искаться в текущем каталоге.
3) FindFirst требует задания атрибута для файла - здесь мы искали все файлы, если надо какие-то определенные (например только скрытые, или только каталоги) то надо это указать, список всех атрибутов я уже приводил выше.
4) SearchRec переменная связывает во едино FindFirst и FindNext, но требует ресурсов для своей работы, поэтому желательно ее освободить после поиска процедурой FindClose(SearchRec) - на самом деле утечки памяти небольшие, но если программа работает в цикле и долгое время пожирание ресурсов будет значительным.
5) FindFirst/FindNext - работают не открывая файлы, поэтому они корректно находят даже Swap файлы Windows...
Поиск файлов по дереву каталогов с заходом в подкаталоги разобран здесь: Проход дерева каталогов
== Работа через WinAPI ==
Раздел написан Podval (примеры к сожалению на С++)
 Любителям WinAPI посвящается...
Функции FileOpen, FileSeek, FileRead.
Возьмем форму, положим на нее кнопку, грид и Опен диалог бокс.
Это для Билдера, но какая нам в данном случае разница?
void __fastcall TForm1::Button1Click(TObject *Sender)
{
 int iFileHandle;
 int iFileLength;
 int iBytesRead;
 char *pszBuffer;
 if (OpenDialog1->Execute()) {
 try {
 iFileHandle = FileOpen(OpenDialog1->FileName, fmOpenRead);
 iFileLength = FileSeek(iFileHandle,0,2);
 FileSeek(iFileHandle,0,0);
 pszBuffer = new char[iFileLength+1];
 iBytesRead = FileRead(iFileHandle, pszBuffer, iFileLength);
 FileClose(iFileHandle);
 for (int i=0;i<iBytesRead;i++) {
 StringGrid1->RowCount += 1;
 StringGrid1->Cells[1][i+1] = pszBuffer[i];
 StringGrid1->Cells[2][i+1] = IntToStr((int)pszBuffer[i]);
 }
 delete [] pszBuffer;
 }
 catch(...) {
 Application->MessageBox("Can't perform one of the following file operations: Open, Seek, Read, Close.", "File Error", IDOK);
 }
 }
}
Потренируемся еще.
Функции FileExists, RenameFile, FileCreate, FileWrite, FileClose.
Бросим на форму Save dialog box.
#include <dir.h>
void __fastcall TForm1::Button1Click(TObject *Sender)
{
 char szFileName[MAXFILE+4];
 int iFileHandle;
 int iLength;
 if (SaveDialog1->Execute()) {
 if (FileExists(SaveDialog1->FileName)) {
 fnsplit(SaveDialog1->FileName.c_str(), 0, 0, szFileName, 0);
 strcat(szFileName, ".BAK");
 RenameFile(SaveDialog1->FileName, szFileName);
 }
 iFileHandle = FileCreate(SaveDialog1->FileName);
 // Write out the number of rows and columns in the grid.
 FileWrite(iFileHandle, (char*)&(StringGrid1->ColCount), sizeof(StringGrid1->ColCount));
 FileWrite(iFileHandle, (char*)&(StringGrid1->RowCount), sizeof(StringGrid1->RowCount));
 for (int x=0;x<StringGrid1->ColCount;x++) {
 for (int y=0;y<StringGrid1->RowCount;y++) {
 // Write out the length of each string, followed by the string itself.
 iLength = StringGrid1->Cells[x][y].Length();
 FileWrite(iFileHandle, (char*)&iLength, sizeof(iLength));
 FileWrite(iFileHandle, StringGrid1->Cells[x][y].c_str(), StringGrid1->Cells[x][y].Length());
 }
 }
 FileClose(iFileHandle);
 }
}
(с) Оба примера взяты из хелпа по Borland C++ Builder 5.
Дата/время файла
extern PACKAGE int fastcall FileGetDate(int Handle);
extern PACKAGE int fastcall FileSetDate(int Handle, int Age);
Первоисточник тот же.
Атрибуты файла.
extern PACKAGE int __fastcall FileGetAttr(const AnsiString FileName);
extern PACKAGE int fastcall FileSetAttr(const AnsiString FileName, int Attr);
Может быть, такая функция полезна будет, хотя о программном поиске файлов и так много наговорено. Но на всякий случай, как говорится.
В этом примере идет поиск в текущем каталоге и каталоге Windows
void __fastcall TForm1::Button1Click(TObject *Sender)
{
 char buffer[256];
 GetWindowsDirectory(buffer, sizeof(buffer));
 AnsiString asFileName = FileSearch(Edit1->Text, GetCurrentDir() + AnsiString(";") + AnsiString(buffer));
 if (asFileName.IsEmpty())
 ShowMessage(AnsiString("Couldn't find ") + Edit1->Text + ".");
 else
 ShowMessage(AnsiString("Found ") + asFileName + ".")
}
В дополнение к Дате/Времени
extern PACKAGE int __fastcall FileAge(const AnsiString FileName);
Для конвертации возвращаемого значения в TDateTime:
extern PACKAGE int fastcall FileDateToDateTime(int FileDate);
Author: Vit 
Source: Взято с Vingrad.ru http://forum.vingrad.ru
ID: 03115
Category:Файловая система
Category:Статьи
Category:Учебники